GNUPlot – Basics via Examples

Saurish Chakrabarty

Acharya Prafulla Chandra College

The topics discussed are taken from the WBSU B.Sc. (NEP) syllabus (First Semester, Paper: PHSDSC101P – Mathematical Methods I Lab)

References

- http://www.gnuplot.info/
- P. K. Janert, "Gnuplot in Action", https://hadron.physics.fsu.edu/ eugenio/comphy/gnuplotbook.pdf

1 Plotting Functions

Try these out.

- `plot cos(x)`
  You should be able to see a plot of $\cos x$.
- `plot cos(x) linewidth 3 linetype 2`
  The linewidth and linetype may be mentioned. Linetype includes color and symbol.
- `test`
  This is a command that can be used to check out the available options.
- `plot [0:2*pi][-2:2] cos(x), cos(x-pi/4) lw 3 lt 8 dashtype 2, cos(x-pi/2) lw 3 lt 7 dt 3`
  Multiple functions can be plotted together using different linewidths, linetypes and dashtypes.
- `plot [-2:2][-.5:.5] x**4-x**2 with points lt 8 pointsize 2 pointtype 7`
  `plot [-2:2][-.5:.5] x**4-x**2 w p lt 8 ps 2, x**3-x w linespoints pt 7 ps 2 lw 2`
  By default, for functions, GNU Plot draws lines connecting the sampled points. Points can be shown by specifying "with points", or in short, just "w p" The lines can be simultaneously shown using "w lp". The point size may also be specified, *e.g.*, "ps 3".
- `p(x)=exp(-x**2/2)/(2*pi)**0.5`
  `w=3`
  `plot [-3:3] p(x) lw w, 2*p(2*x) lw w dt w, p(x/2)/2 lw w dt 4`
  The first line defines a standard normal (Gaussian) probability density function. Using this, some plots are made. Note that the line width is chosen to be a variable.
- `f(x)=x<-1?0:(x<0?1+x:(x<1?1-x:0))`
  `p [-2:2] f(x) lw 2, 2*f(2*x) dt 2 lw 3, f(x-0.7) dt 4 lw 2`
  The first line defines a simple triangular pulse centered around the origin. Using this, some plots are made in the next line.
- `p(x)=exp(-(x-m/s)**2/2)/s/(2*pi)**0.5`
  `plot [-3:10] m=0 s=1 p(x), m=4 s=2 p(x), m=8 p(x)`
  See what this does.

As seen at various instances above, GNU Plot can guess word completions from context.

2 Modifying the Appearance of Graphs

- `set zeroaxis`
  Show the axes.
- `plot [-1:3][-0.5:1] 1-exp(-x) title 'growth', x-x**2/2 title 'model'`
  Change the labels shown in the legend.
- `set size square`
  This is to make a plot with 1:1 aspect ratio.
- `set samples 1000`
  This is to set the number of samples (data points) used for plotting (default: 100).
- `set xlabel 'Radius'`
  `set ylabel 'Volume'`
  To set axis labels.
- `set title 'Volume vs Radius'`
  To set the title.
- `set term wxt font ",20"`
  This is to set the global font size to 20.
- `set title 'Volume vs Radius' font ",30"`
  To set the font size of just the title (or for any one field).
- `set logscale`
  For logarithmic axes.
- `set logscale x`
  To set logarithmic x-axis.
- `unset logscale`
  To turn off logarithmic axes.

3 Set, Unset, Reset, Load, Save, Replot

- "show": see the current value of an option.
  "set": set value.
  "unset": turn of the set value of an option.
  "reset": restore everything to default values except "terminal" and "output".
- "load $file.gp$": load a file named $file.gp$.
- "save $file.gp$": save current commands into a file named $file.gp$.
- "replot": repeat last plot.

4 Exporting Plots

- `set terminal pdfcairo`
  `set output "graph.pdf"`
  Setting pdf terminal and the name of the output file.
- `set output`
  After the plotting commands, set an empty output to flush the plots to the pdf file. This is not needed if you exit GNU Plot after plotting.
- `set terminal wxt` or `set terminal qt` (depending on what your default terminal is)
  `set output`
  Reset to the usual terminal.

5 Polar Plots

- `set polar`
  Setting polar plot. The independent variable, θ, is denoted by t.
- `plot 1`
  Circle.
- `plot 1/(1+0.5*cos(t))`
  Ellipse.
- `set angles degrees`
  If needed. Default: radians.
- `set size square`
  `set tics` (or "rtics")
  `set grid polar`
  `set grid polar lt 8 lw 0.4`
  Useful settings for polar plots.

6 Parametric Plots

- `set parametric`
  Setting parametric plot. The parameter is denoted by t.
- `p [0:2*pi] cos(t),sin(t), 1.2*cos(t),0.8*sin(t)`
  A circle and an ellipse.
- `p [0:2*pi] cos(5*t),0.8*sin(3*t)`
  Lissajous figure.

7 Importing Multicolumn Data

In a linux terminal, run the following command.

```
awk 'BEGIN{for(i=0;i<100;i++)print i,exp(-0.05*i)+(rand()-0.5)/10,
                                    0.01*i+(rand()-0.5)*0.05}' > xy.dat
```

Alternatively, use your favorite method to create a file with many rows and at least three columns of numbers.

- `p "xy.dat"`
- `p "xy.dat" u 1:3`
- `p "xy.dat" u 1:cos($3)`
- `p "xy.dat" u 1:(cos($3/10)) w l, "xy.dat" u 1:(sin($3/10)) w l`

8 Fitting Data

```
f(x,m,c)=m*x+c
fit f(x,m,c) "xy.dat" u 1:3 via m,c
p "xy.dat" u 1:3, f(x,m,c)
print m,c
```

9 Surface Plots

- `splot f(x,y)`
- `set isosamples 50`
  Use a 50×50 grid instead of 10×10 (default).
- `splot [-2:2][-2:2] exp(-x**2-y**2)*cos(5*x*y)`
- `set view 70,30,0.5,0.8`
  View from $\theta = 70°$, $\phi = 30°$, $xy\_scale = 0.5$, $z\_scale = 0.8$.
- `set xyplane 0.5`
  xy-plane at $z = 0.5$.

10 Color Plots

- `unset surface`
  `set pm3d`
  Unset surface and turn on the pm3d (color) feature.
- `splot f(x,y)`

11 Contour Plots

- `set contour`
  Turn on contours.
- `splot f(x,y)`
- `set contour base` (default)
  `set contour surface`
  `set contour both`
  Three common settings.
- `unset surface`
  `set view map`
  `set contour`
  For planar contour plot.

12 Simple "Movie"

```
do for [t=0:100] {
plot sin(x-t)
pause 0.1
}
```

13 Help Command

For any topic, try `help topic`.

Last updated: September 22, 2024